


durexForth

Operators Manual



Contents

1	Intr	roduction	4						
	1.1	Forth, the Language	4						
		1.1.1 Why Forth?	4						
		1.1.2 Comparing to other Forths	4						
	1.2	Helpdesk	5						
2	Tut	orial (6						
	2.1	Interpreter	6						
	2.2	Editor	6						
	2.3								
	2.4								
	2.5		3						
	2.6		3						
			3						
			9						
	2.7	How to Learn More	9						
			9						
		2.7.2 Other	9						
3	Edi	tor 10	0						
	3.1	Key Presses)						
		3.1.1 Inserting Text)						
		3.1.2 Navigation)						
		3.1.3 Saving & Quitting	1						
		3.1.4 Text Manipulation	1						
4	Forth Words 12								
	4.1	Stack Manipulation	2						
	4.2	Utility	3						
	4.3	Mathematics	3						
	4.4	Signed Mathematics							
	4.5	Logic	4						
	4.6	Memory	4						
	4.7	Compiling	5						
	4.8	Variables							
		4.8.1 Values							
		4.8.2 Variables							
		183 Arrays							

	4.9	Control Flow		16
	4.10	Keyboard Input		17
		Editing		17
	4.12	2 Strings		17
		3 Vectored Execution		17
	4.14	Debugging		18
		5 System State		18
		6 Disk I/O		18
		Kernel Calls		18
		3 "Missing" Words		18
5	Gra	aphics		19
	5.1	Turtle Graphics		19
	5.2			19
A	Asse	sembler Mnemonics		21
\mathbf{B}	Mer	emory Map	:	22

Introduction

1.1 Forth, the Language

1.1.1 Why Forth?

Forth is a different language. It is old, a little weird and makes you think different.

What is cool about it? It is a very low-level and minimal language that has a few rough edges. At the same time, it is easy to make it a very high-level and domain-specific language, much like Lisp.

Compared to C64 Basic, Forth is more attractive in almost every way. It is a lot more fast, memory effective and powerful.

Compared to C, specifically cc65, the story is a little different. It's hard to make a fair comparison. Theoretically Forth code can be very memory efficient, and it's possible to make Forth code that is leaner than C code. But it is also true that cc65 code is generally faster than Forth code.

The main advantage of Forth is that the environment runs on the actual machine. It would not be a lot of fun to use a C compiler that runs on a standard C64. But with Forth, it's possible to create an entire development suite with editor, compiler and assembler that runs entirely on the C64.

Another advantage is that Forth has an interpreter. Compared to cross-compiling, it is really nice to make small edits and tweaks without going through the entire edit-compile-link-transfer-boot-run cycle.

For a Forth introduction, please refer to the excellent Starting Forth by Leo Brodie. As a follow-up, I recommend Thinking Forth by the same author.

1.1.2 Comparing to other Forths

There are other Forths for c64, most notably Blazin' Forth. Blazin' Forth is excellent, but durexForth has some advantages:

- Store your Forth sources as text files no crazy block file system.
- durexForth is smaller.
- The editor is a vi clone.
- durexForth is open source (available at Google Code).

1.2 Helpdesk

If you run into problems or have ideas for improvement, feel free to contact durexForth helpdesk.

Tutorial

2.1 Interpreter

Start up durexForth. If loaded successfully, it will greet you with a friendly ok. You have landed in the interpreter!

Let's warm it up a little. Enter 1 (followed by return). You have now put a digit on the stack. This can be verified by the command .s, which will print out the stack. Now enter . to pop the digit and print it to screen, followed by .s to verify that the stack is empty.

Now some arithmetics. 1000 a * . will calculate $$a \times 1000 and print the result on the screen. 6502 100 / 1- . will calculate and print (\$6502/\$100) - 1. Let's define a word bg for setting the border color...

```
: bg d020 c!;
```

 $0~bg,\,1~bg$ and so on will let you set border color. Cool! Now let's head on to making our first "real" program...

2.2 Editor

The editor (fully described in chapter 3) is convenient for editing larger pieces of code. With it, you keep an entire source file loaded in RAM, and you can recompile and test it easily.

Start the editor by typing vi. You will enter the pink editor screen.

To enter text, first press i to enter insert mode. This mode allows you to insert text into the buffer. You can see that it's active on the I that appears in the lower left corner.

This is a good start for making a program. But first, let's get rid of the junk we created in the last section. Enter:

forget bg

 \dots and press \leftarrow to leave insert mode. The line you entered forgets the bg word that you defined in the last section, and everything defined after it. Let's try out if it works.

First, quit the editor by pressing :q. You should now be back in the interpreter screen. Verify that the word bg still exists by entering 0 bg, 1 bg like you did before. Then, jump back to the editor using the command fg (foreground). You should return to your edit buffer with the lonely forget bg line.

Now, compile and run the buffer by pressing F7. You will be thrown out to the interpreter again. Entering bg should now give you the error bg?. Success—we have forgotten the bg word. Now, get back into the editor by entering fg. Under forget bg, add the following lines:

```
: flash d020 c@ 1+ d020 c! recurse ; flash
```

flash will cycle the border color infinitely. Before trying it out, go up and change forget bg to forget flash. This makes sure you won't run out of RAM, no matter how many times you recompile the program. Now press F7 to compile and run. If everything is entered right, you will be facing a wonderful color cycle.

To get back into the editor, press Restore key. Let's see how we can factor the program to get something more Forth'y:

```
forget bg
: bg d020 ; # border color addr
: inc dup c@ 1+ swap c! ; ( addr -- )
: flash dup inc recurse ; ( addr -- )
bg flash
```

(Note: Parentheses are used for multi-line comments or describing arguments and return values. # is used for single-line comments.)

Of course, it is a matter of taste which version you prefer. Press F7 to see if the new version runs faster or slower.

2.3 Assembler

If you need to flash as fast as possible, use the assembler:

```
:asm flash
here @ # push curr addr
d020 inc,
jmp, # jmp to pushed addr
;asm
flash
```

:asm and ;asm define a code word, just like : and ; define Forth words. Within a code word, you can use assembler mnemonics.

Note: As the x register contains the durexForth stack depth, it is important that it remains unchanged at the end of the code word.

2.4 Console I/O Example

This piece of code reads from keyboard and sends back the chars to screen:

```
: init 0 linebuf c! ; # disable key buffering
: foo init begin key emit again ;
foo
```

2.5 Avoiding Stack Crashes

durexForth should be one of the fastest and leanest Forths for the C64. To achieve this, there are not too many niceties for beginners. For example, compiled code has no checks for stack overflow and underflow. This means that the system may crash if you do too many pops or pushes. This is not much of a problem for an experienced Forth programmer, but until you reach that stage, handle the stack with care.

One technique that can be useful to avoid stack crashes is to add comments about what your stack looks like at different times. In this example, we imagine a graphics word "drawbox" that draws a black box. (--) indicates that it takes no arguments on stack, and on exit it should leave nothing on the stack. The comments inside the word indicate what the stack looks at the given points of time.

```
: drawbox ( color -- )
10 begin dup 20 < while # color x
10 begin dup 20 < while # color x y
2dup # color x y x y
4 pick # color x y x y color
blkcol # color x y
1+ repeat drop # color x
1+ repeat 2drop;</pre>
```

Once the word is working, it may be nice to again remove the # comments as they are no longer that interesting to read.

2.6 Configuring durexForth

2.6.1 Stripping Modules

By default, durexForth boots up with all modules pre-compiled in RAM:

```
debug Words for debugging.
```

```
asm The assembler.
```

vi The text editor.

ls List disk contents.

gfx Graphics module.

To reduce RAM usage, you may make a stripped-down version of durexForth. Do this by following these steps:

- 1. Issue forget modules to forget all modules.
- 2. Optionally re-add the modules marker with: modules;
- 3. One by one, load the modules you want included with your new Forth. (E.g. s" debug" load)
- 4. Save the new system with e.g. s" acmeforth" save-forth.

2.6.2 Custom Start-Up

You may launch a word automatically at start-up by setting the variable start to the execution token of the word. Example: loc megademo >cfa start!

To save the new configuration to disk, use save-forth.

2.7 How to Learn More

2.7.1 Internet Resources

Books and Papers

- Starting Forth
- Thinking Forth
- Moving Forth: a series on writing Forth kernels
- Blazin' Forth An inside look at the Blazin' Forth compiler
- The Evolution of FORTH, an unusual language
- A Beginner's Guide to Forth

Other Forths

- colorForth
- JONESFORTH
- colorForthRay.info How_to: with Ray St. Marie

2.7.2 Other

• durexForth source code

Editor

The editor is a vi clone. Launch it by entering s" foo" vi in the interpreter (foo being the file you want to edit). You may also enter vi with no parameters on stack - in that case, it will create a text file named "untitled". For more info about vi style editing, see the Vim web site.

The position of the editor buffer is controlled by the variable bufstart. The default address is \$5000.

3.1 Key Presses

3.1.1 Inserting Text

Following commands enter insert mode. Insert mode allows you to insert text. It can be exited by pressing \leftarrow .

- i Insert text.
- a Append text.
- o Open new line after cursor line.
- O Open new line on cursor line.
- cw Change word.

3.1.2 Navigation

hjkl Cursor left, down, up, right.

Cursor Keys ...also work fine.

- U Half page up.
- ${\bf D}\,$ Half page down.
- ${\bf b}\,$ Go to previous word.
- w Go to next word.
- **0** Go to line start.

- \$ Go to line end.
- g Go to start of file.
- ${f G}$ Go to end of file.

3.1.3 Saving & Quitting

After quitting, the editor can be re-opened with Forth command fg, and it will resume operations with the edit buffer preserved.

ZZ Save and exit.

- :q Exit.
- :w Save. (Must be followed by return.)
- :w!filename Save as.

 ${\bf F7}$ Compile and run editor contents. Press Restore key to return to editor.

3.1.4 Text Manipulation

- ${f r}$ Replace character under cursor.
- ${f x}$ Delete character.
- X Backspace-delete character.
- dw Delete word.
- dd Cut line.
- yy Yank (copy) line.
- **p** Paste line below cursor position.
- ${\bf P}\,$ Paste line on cursor position.
- J Join lines.

Forth Words

4.1 Stack Manipulation

```
drop (\mathbf{a} – ) Drop top of stack.
dup (a - a a) Duplicate top of stack.
\mathbf{swap} ( \mathbf{a}\ \mathbf{b} - \mathbf{b}\ \mathbf{a} ) Swap top stack elements.
over (a b - a b a) Make a copy of the second item and push it on top.
\mathbf{rot} ( \mathbf{a} \ \mathbf{b} \ \mathbf{c} - \mathbf{b} \ \mathbf{c} \ \mathbf{a} ) Rotate the third item to the top.
-rot ( \mathbf{a} \ \mathbf{b} \ \mathbf{c} - \mathbf{c} \ \mathbf{a} \ \mathbf{b} ) rot rot
2drop ( a b - ) Drop two topmost stack elements.
2dup ( a b - a b a b ) Duplicate two topmost stack elements.
2<br/>swap ( a b c d – c d a b ) Swap topmost double stack elements.
?dup ( \mathbf{a} - \mathbf{a} \ \mathbf{a}? ) Dup a if a differs from 0.
nip (a b - b) swap drop
tuck (ab - bab) dup -rot
pick ( x_u ... x_1 x_0 u – x_u ... x_1 x_0 x_u ) Pick from stack element with depth
       u to top of stack.
>r ( a - ) Move value from top of parameter stack to top of return stack.
r > (a - ) Move value from top of return stack to top of parameter stack.
\mathbf{r} ( -\mathbf{a} ) Fetch value of top of return stack (without lifting it).
rdrop ( – ) Drop value on top of return stack.
```

4.2 Utility

- . (a) Print top value of stack.
- **c.** (\mathbf{a}) Print top byte of stack.
- .s See stack contents.
- emit (\mathbf{a}) Print top byte of stack as a PETSCII character.
- # Comment to end of line.
- (Start multi-line comment.
-) End multi-line comment.

4.3 Mathematics

These words assume that the lowest number is 0 and highest is FFFF.

- 1+(a-b) Increase top of stack value by 1.
- 1- ($\mathbf{a} \mathbf{b}$) Decrease top of stack value by 1.
- 2+ (a-b) Increase top of stack value by 2.
- 2* (a b) Fast multiply top of stack value by 2.
- 2/(a-b) Fast divide top of stack value by 2.
- 100/ (a b) Divides top of stack value by \$100.
- +! (n a) Add n to memory address <math>a.
- + (a b c) Add a and b.
- ($\mathbf{a} \ \mathbf{b} \mathbf{c}$) Subtract b from a.
- * ($\mathbf{a} \ \mathbf{b} \mathbf{c}$) Multiply a with b.
- $\mathbf{d^*}$ ($\mathbf{a}~\mathbf{b}-\mathbf{msw}~\mathbf{lsw}$) 32-bit multiply a with b.
- /mod (a b r q) Divide a with b. r = rest, q = quotient.
- / ($\mathbf{a} \ \mathbf{b} \mathbf{q}$) Divide a with \mathbf{b} .
- \mathbf{mod} ($\mathbf{a}\ \mathbf{b} \mathbf{r}$) Rest of a divided by b.
- < (a b c) Is a less than b?
- > (a b c) Is a greater than b?
- >= (a b c) Is a greater than or equal to b?
- <= (a b c) Is a less than or equal to b?

4.4 Signed Mathematics

These words treat numbers like they are signed, meaning that numbers 8000 - FFFF are negative and 0 - 7FFF are positive. (E.g., FFFF means -1.)

```
0< (a - b) Is a negative?</li>
negate (a - b) Negates a.
abs (a - b) Gives absolute value of a.
s< (ab - c) Is a less than b? (Signed comparison.)</li>
s> (ab - c) Is a greater than b? (Signed comparison.)
```

4.5 Logic

```
    0= (a - b) Is a equal to zero?
    = (ab - c) Is a equal to b?
    (ab - c) Does a differ from b?
    and (ab - c) Binary and.
    or (ab - c) Binary or.
    xor (ab - c) Binary exclusive or.
    not (a - b) Flip all bits of a.
```

4.6 Memory

```
! ( value address - ) Store 16-bit value at address.
@ ( address - value ) Fetch 16-bit value from address.
c! ( value address - ) Store 8-bit value at address.
c@ ( address - value ) Fetch 8-bit value from address.
fill ( byte addr len - ) Fill range [addr, len + addr) with byte value.
cmove ( src dst len - ) Copy len bytes from src to dst. The move begins with the contents of src and proceeds towards high memory.
cmove> ( src dst len - ) Byte-to-byte copy like cmove, but starts with ad-
```

forget xxx Forget Forth word xxx and everything defined after it.

dress src + len - 1 and proceeds towards src.

4.7 Compiling

- : Start compiling Forth word at here position.
- ; End compiling.
- , (n-) Write word on stack to here position and increase here by 2.
- \mathbf{c} , (\mathbf{n}) Write byte on stack to here position and increase here by 1.
- **literal** (n -) Compile a value from the stack as a literal value.
- [() Leave compile mode. Execute the following words immediately instead of compiling them.
-] () Return to compile mode.

immed Mark the word being compiled as immediate (i.e. inside colon definitions, it will be executed immediately instead of compiled).

[compile] xxx Compile the immediate word xxx instead of executing it.

['] xxx Compile the execution token of word xxx as a literal value.

create xxx Create a dictionary header with name xxx.

4.8 Variables

4.8.1 Values

Values are fast to read, slow to write.

- : foo 1; Define value foo.
- 1 value foo Equivalent to the above.

foo Fetch value of foo.

0 to foo Set foo to 0.

4.8.2 Variables

Variables are faster to write to than values.

var foo Define variable foo.

foo @ Fetch value of foo.

1 foo! Set value of foo to 1.

4.8.3 Arrays

10 allot value foo Allocate 10 bytes to array foo.

1 foo 2 +! Store 1 in position 2 of foo.

foo dump See contents of foo.

It is also possible to build arrays using create. The initialization is easier, but access is slightly different:

```
create 2powtable
1 c, 2 c, 4 c, 8 c,
10 c, 20 c, 40 c, 80 c,
: 2pow ( n -- 2**n ) ['] 2powtable + c@;
```

4.9 Control Flow

Control functions only work in compile mode, not in interpreter.

if ... then condition IF true-part THEN rest

if ... else ... then condition IF true-part ELSE false-part THEN rest

begin ... again Infinite loop.

begin ... until BEGIN loop-part condition UNTIL.

Loop until condition is true.

begin ... while ... repeat BEGIN condition WHILE loop-part REPEAT.

Repeat loop-part while condition is true.

exit Exit function.

recurse Jump to the start of the word being compiled.

case ... endcase, of ... endof Switch statements.

```
: tellno ( n -- )
case
1 of ." one" endof
2 of ." two" endof
3 of ." three" endof
." other"
endcase
```

4.10 Keyboard Input

- $key\ (-n\)$ Reads a character from input. Buffered/unbuffered reading is controlled by the <code>linebuf</code> variable.
- ${f word}$ ($-{f addr}$) Reads a word from input and put the string address on the stack.
- interpret (value) Interprets a word from input and puts it on the stack.
- linebuf This variable switches between buffered/unbuffered input. Disable
 input buffering with 0 linebuf c!, enable with 1 linebuf c!.

4.11 Editing

- vi(-) Opens editor with untitled buffer.
- vi (file) Opens file in editor. Try s" foo" vi.
- fg Re-open editor to pick up where it left.

4.12 Strings

- ." Print a string. E.g. . " foo"
- s" Run time: (strptr strlen) Compile time: (str)

Define a string and put it on the stack. E.g. s" foo".

In run time, it puts the string length and a text pointer on the stack. In compile time (inside a colon definition), it puts a pointer to a Pascal-string on the stack.

4.13 Vectored Execution

- ' xxx (-addr) Compile-time only: Find execution token of word xxx.
- lit xxx (-addr) Equal to 'but used for clarity. Use 'lit , , to compile the (run-time) value on top of stack.
- \mathbf{exec} (\mathbf{xt}) Execute the execution token on top of stack.
- loc xxx (addr) Run-time only: Get adress of word xxx.
- >cfa (addr xt) Get execution token (a.k.a. code field adress) of word at adress addr.
 - Example: f = loc f > cfa exec

4.14 Debugging

Debugging words are loaded with s" debug" load.

words List all defined words.

sizes List sizes of all defined words.

 $\operatorname{dump} (n -)$ Memory dump starting at address n.

n Continue memory dump where last one stopped.

see word Decompile Forth word and print to screen. Try see see.

4.15 System State

latest (variable) Position of latest defined word.

here (variable) Write position of the Forth compiler (usually first unused byte of memory). Many C64 assemblers refer to this as program counter or *.

sp@ (-addr) Address of stack top before sp@ is executed.

sp0 (value) Address of stack bottom.

blink (status -) Disable/enable cursor blink. (0 = off, 1 = on)

4.16 Disk I/O

load (filenameer filenamelength –) Load and execute/compile file.

loadb (filenameetr filenamelength dst –) Load binary block to dst.

saveb (start end filenameptr filenamelength –) Save binary block.

scratch (filenameptr filenamelength –) Scratch file.

4.17 Kernel Calls

Safe kernel calls may be done from Forth words using jsr (addr –). The helper variables ar, xr and yr can be used to set arguments and get results through the a, x and y registers.

Example: 30 ar ! ffd2 jsr prints 0 on screen.

4.18 "Missing" Words

The following words might be expected in a "normal" Forth, but are not included in durexForth for the sake of keeping it lean:

- do ... loop, i, j
- */, */mod

Graphics

As of durexForth v1.2, high-resolution graphics support is included.

5.1 Turtle Graphics

Turtle graphics are mostly known from LOGO, a 1970s programming language. It enables control of a turtle that can move and turn while holding a pen. The turtle graphics library is loaded with s" turtle" load.

```
init ( – ) Initializes turtle graphics.
```

forward (px -) Moves the turtle px pixels forward.

back (px -) Moves the turtle px pixels back.

left (deg -) Rotates the turtle deg degrees left.

right (deg -) Rotates the turtle deg degrees right.

penup (–) Pen up (disables drawing).

pendown (-) Pen down (enables drawing).

5.2 High-Resolution Graphics

The high-resolution graphics library is loaded with s" gfx" load. It is inspired by "Step-by-Step Programming Commodore 64: Graphics Book 3." Some demonstrations can be found in gfxdemo.

hires (–) Enters the high-resolution drawing mode.

lores (–) Switches back to low-resolution text mode.

clrcol (colors –) Clears the high-resolution display using colors. Colors is a byte value with foreground color in high nibble, background color in low nibble. E.g. 15 clrcol clears the screen with green background, white foreground.

blkcol (col row colors –) Changes colors of the 8x8 block at given position.

```
plot (x y -) Sets the pixel at x, y.
peek ( \mathbf{x} \ \mathbf{y} - \mathbf{p} ) Gets the pixel at x, y.
line ( \mathbf{x} \ \mathbf{y} - \mathbf{)} Draws a line to \mathbf{x}, \mathbf{y}.
circle ( x y r - ) Draws a circle with radius r around x, y.
{f erase} ( {f mode} – ) Changes blit method for line drawing. 1 erase uses {f xor} for
      line drawing, O erase switches back to or.
paint (x y -) Paints the area at x, y.
text (column row str strlen - ) Draws a text string at the given position.
      E.g. 10 8 s" hallo" text draws the message "hallo" at column 16, row
drawchar ( column row char – ) Draws a custom character at given column
      and row.
defchar Defines an 8x8 character to use with the drawchar word. Example:
      defchar sqr
      00000000
      00000000
      00111100
      00111100
      00111100
      00111100
      00000000
      00000000
      2 2 sqr drawchar
```

...draws a square at column 2, row 2.

Appendix A

Assembler Mnemonics

adc,#	bvs,	eor,(x)	lsra,	sbc,#
adc,	clc,	eor,(y)	lsr,	sbc,
adc,x	cld,		lsr,x	sbc,x
adc,y	cli,	inc,		sbc,y
adc,(x)	clv,	inc,x	nop,	sbc,(x)
adc,(y)				sbc,(y)
	cmp,#	inx,	ora,#	
and,#	cmp,	iny,	ora,	sec,
and,	cmp,x		ora,x	sed,
and,x	cmp,y	jmp,	ora,y	sei,
and,y	cmp,(x)	<pre>jmp,()</pre>	ora,(x)	
and, (x)	cmp,(y)		ora,(y)	sta,
and,(y)		jsr,		sta, x
	cpx,#		pha,	sta,y
asla,	cpx,	lda,#	php,	sta,(x)
asl,		lda,	pla,	sta,(y)
asl,x	cpy,#	lda,x	plp,	
	cpy,	lda,y		stx,
bcc,		lda,(x)	rola,	stx,y
bcs,	dec,	lda,(y)	rol,	
beq,	dec,x		rol,x	sty,
		ldx,#		sty,x
bit,	dex,	ldx,	rora,	
	dey,	ldx,y	ror,	tax,
bmi,			ror,x	tay,
bne,	eor,#	ldy,#		tsx,
bpl,	eor,	ldy,	rti,	txa,
brk,	eor,x	ldy,x	rts,	txs,
bvc,	eor,y			tya,

Appendix B

Memory Map

```
... - $83 Parameter stack (grows downwards).
$84 - $89 Temporary registers.
$8a - $8b Instruction pointer.
...
$801 - here Forth Kernel followed by dictionary.
...
bufstart - eof Editor space.
```